Methodology, Modelling, and Consulting Skills
School of Mathematics
The University of Edinburgh
Year: 2022/2023

THE UNIVERSITY of EDINBURGH

# Computer Session 2: Multi-period models and inventory

Let us look at a production planning problem and how to model it in Xpress.

## 1 The basic problem

Let us start with a basic variant of the problem!

> Sailco Corporation must decide how many sailboats to produce in each quarter. The demand is different in each quarter depending on the season. Sailco has no warehouse and therefore all the sailboats must be sold in the same quarter they are produced. Find a production plan that meets the demand of the whole year at minimum cost.

A linear programming model for this problem is:

$$\begin{array}{ll} \text{Min.} & 400x_1 + 400x_2 + 400x_3 + 400x_4 \\ \text{s.t.} & x_1 \geq 40, \\ & x_2 \geq 60, \\ & x_3 \geq 75, \\ & x_4 \geq 25, \\ & x_1, x_2, x_3, x_4 \geq 0. \end{array}$$

Here, $x_t$ is the production in quarter $t$ with ($t = 1$ corresponding to Spring etc).

As we have seen in the previous week, first we need to define indices and structures in the declarations block:

```
declarations
        number_of_periods = 4
        periods = 1..number_of_periods
        period_names: array(periods) of string
        demand: array(periods) of real
        cost: real
        make: array(periods) of mpvar
end-declarations
```

The set `periods` is the set of indices of size `number_of_periods` and the names of these periods are stored in array `period_names`. The meaning of the rest of the structures is obvious. We create file `sailboat_basic.dat` with the following information:

Table 1: Demand data.

| Quarter | Spring | Summer | Autumn | Winter |
|---|---|---|---|---|
| Demand | 40 | 60 | 75 | 25 |

```
period_names: ["Spring" "Summer" "Autumn" "Winter"]
cost: 400
demand: [40 60 75 25]
```

Then, we need to read this information from the file into the structures we have created:

```
initialisations from "sailboat_basic.dat"
        period_names demand cost
end-initialisations
```

Defining the objective function and the constraints is trivial:

```
! Objective function
total_cost:= sum(t in periods) cost*make(t)

! Constraints
forall(t in periods) make(t) >= demand(t)
```

After adding some lines that show the solution, the code is complete; see the file `sailboat_basic.mos` for the full code.

## 2 Multi-period with inventory

The first model we solved is unrealistic, typically we can *store* products. Let us model this!

> Sailco Corporation must decide how many sailboats to produce in each quarter. The demand is different in each quarter depending on the season. **Due to workload limitations, Sailco cannot produce more than 60 sailboats per quarter. Sailco has developed a sufficiently large warehouse where they can store as many sailboats as required at no additional cost.**
>
> Find a production plan that meets the demand of the whole year at the minimum possible cost
>
> Assume, for simplicity, that the sailboats produced in one quarter can be used to meet the demand in that quarter.

The core idea is to introduce two different sets of variables. We let $y_t$ be the number of units *available* for supply in quarter $t$, and let $x_t$ be the number of units *produced* in quarter $t$. So, we have to satisfy a minimum demand, which is a constraint on the $y_t$ and we have a production limit, which is a constraint on the $x_t$.

To couple these variables, we need to model the inventory balance:

$$\begin{aligned}\text{Inventory at end of quarter } t &= \text{Inventory at end of quarter } t-1\\ &\quad + \text{quarter } t \text{ production}\\ &\quad - \text{quarter } t \text{ sellings.}\end{aligned}$$

Let us define, $i_t$ as the inventory level at end of quarter $t$, then we obtain the constraints

$$\begin{aligned} i_1 &= (i_0) + x_1 - y_1,\\ i_t &= i_{t-1} + x_t - y_t,\ t = 2, 3, 4.\end{aligned}$$

The full model is now given by

$$\begin{array}{lll} \text{Min.} & 400x_1 + 400x_2 + 400x_3 + 400x_4 & \\ \text{s.t.} & x_t \leq 60, & t = 1, 2, 3, 4,\\ & i_1 = x_1 - y_1, & \\ & i_t = i_{t-1} + x_t - y_t, & t = 2, 3, 4,\\ & y_1 \geq 40, & \\ & y_2 \geq 60, & \\ & y_3 \geq 75, & \\ & y_4 \geq 25, & \\ & x_t, y_t, i_t \geq 0, & t = 1, 2, 3, 4.\end{array}$$

To model this in Xpress, we need to expand the declarations block to add the new variables and parameters:

```
declarations
        number_of_periods = 4
        periods = 1..number_of_periods
        period_names: array(periods) of string
        demand: array(periods) of real
        cost,production_limit: real
        make,sell,inventory: array(periods) of mpvar
end-declarations
```

Constraints for demand requirements and production limits are straightforward to write:

```
! Production limit
forall(t in periods) make(t) <= production_limit

! Demand satisfaction
forall(t in periods) sell(t) >= demand(t)
```

For the inventory level constraints, we have a different constraint for the first quarter. We will use the `if` command to produce different constraints inside the forall loop.

```
forall(t in periods) do
        if (t>1) then
                inventory(t) = inventory(t-1) + make(t) - sell(t)
```

```
        else
                inventory(1) = make(1) - sell(1)
        end-if
end-do
```

See file `sailboat_inventory.mos` to see the whole code.

## 3 Multi-period with inventory and holding costs

> Sailco Corporation must decide how many sailboats to produce in each quarter. The demand is different in each quarter depending on the season. Due to workload limitations, Sailco cannot produce more than 60 sailboats per quarter. Sailco has developed a sufficiently large warehouse where they can store as many sailboats as required **at a cost of £50 per unit.**
>
> Find a production plan that meets the demand of the whole year at the minimum possible cost
>
> Assume, for simplicity, that the sailboats produced in one quarter can be used to meet the demand in that quarter.

For this modification, we have to introduce the inventory variables $i_t$ into the objective function. The full model is given by

$$
\begin{array}{lll}
\text{Min.} & 400x_1 + 400x_2 + 400x_3 + 400x_4 + & \\
 & \quad +50i_1 + 50i_2 + 50i_3 + 50i_4 & \\
 & & \\
\text{s.t.} & x_t \leq 60, & t = 1,2,3,4, \\
 & i_1 = x_1 - y_1, & \\
 & i_t = i_{t-1} + x_t - y_t, & t = 2,3,4, \\
 & y_1 \geq 40, & \\
 & y_2 \geq 60, & \\
 & y_3 \geq 75, & \\
 & y_4 \geq 25, & \\
 & x_t, y_t, i_t \geq 0, & t = 1,2,3,4.
\end{array}
$$

In order to consider holding costs in the previous model, we include a new parameter `holding_cost` which is read from the data file `sailboat_inventory_holding.dat` and we add the expression `sum(t in periods) holding_cost*inventory(t)` to the objective function. You can see the code in file `sailboat_inventory_holding.mos`.

## 4 Multi-product

Let us extend this model to multiple products. Sailco Corporation must decide how many *sailboats and surfboards* to produce at each quarter. The demand is given in Table 2. Every sailboat and surfboard takes a number of hours to complete and costs a certain amount of money to produce (see Table 3). The company has 1860 h of work available per quarter. The

Table 2: Demand for sailboats and surfboards.

| | Spring | Summer | Autumn | Winter |
|---|---|---|---|---|
| Sailboat | 40 | 60 | 75 | 25 |
| Surfboard | 190 | 350 | 130 | 20 |

Table 3: Resources and costs.

| | Work (h) | Storage cost (£) | Production cost (£) |
|---|---|---|---|
| Sailboat | 20 | 50 | 400 |
| Surfboard | 3 | 2 | 35 |

company has a warehouse for sailboats and surfboards with different storage costs for each (see also Table 3). What is the production plan that meets the demand of the whole year at minimum cost?

To model multiple products, we have to copy our variables from the preceding examples. We do this be adding an index $p$ for the product, we are producing. So, now we have variables $x_{pt}$ for the amount of product $p$ we produce in quarter $t$. Similarly, we introduce the variables $y_{pt}$ for the number of product $p$ available for supply in quarter $t$, and $i_{pt}$ for the inventory on hand of of product $p$ at the end of quarter $t$.

Since we now have two-index variables, some of the arrays need to re-dimensioned. For example, demand is now a two-dimensional matrix:

```
declarations
        number_of_periods = 4
        periods = 1..number_of_periods
        period_names: array(periods) of string
        number_of_products = 2
        products = 1..number_of_products
        product_names: array(products) of string
        demand: array(products,periods) of real
        cost, holding_cost, hours_needed: array(products) of real
        production_limit:real
        make,sell,inventory: array(products,periods) of mpvar
end-declarations
```

In the objective function we have now double sums:

Note that it is the same to write either

```
sum(p in products, t in periods)
```

or

```
sum(p in products) sum(t in periods)
```

The constraints for the limit on working hours per quarter can be easily written using a loop:

```
! Production limit
forall(t in periods) sum(p in products)( hours_needed(p)*make(p,t)) <= production_limit
```

Now, the constraints for meeting the demand are written using a two-index loop:

```
! Demand satisfaction
forall(p in products, t in periods) sell(p,t) >= demand(p,t)
```

And the same happens with the inventory balance constraints (although we need to write these lines carefully because there is an `if` command that checks one of the two indices):

```
! Inventory balance
forall(p in products, t in periods) do
        if (t>1) then
                inventory(p,t) = inventory(p,t-1) + make(p,t) - sell(p,t)
        else
                inventory(p,1) = make(p,1) - sell(p,1)
        end-if
end-do
```

Finally, we would like to show the information of the solution on the screen. You can find in the file `sailboat_inventory_multiproduct.mos` how this is done. Could you say what is the `if` for?